

varlmar — LM test for residual autocorrelation after var or svar

Description	Quick start	Menu	Syntax
Options	Remarks and examples	Stored results	Methods and formulas
References	Also see		

Description

`varlmar` implements a Lagrange multiplier (LM) test for autocorrelation in the residuals of VAR models, which was presented in [Johansen \(1995\)](#).

Quick start

Test the null hypothesis of no autocorrelation for the first two lags of the residuals after `var` or `svar`

```
varlmar
```

Same as above, but test the first 5 lags

```
varlmar, mlag(5)
```

Perform test for the first two lags of the residuals of a vector autoregression using estimation results stored in `myest`

```
varlmar, estimates(myest)
```

Menu

Statistics > Multivariate time series > VAR diagnostics and tests > LM test for residual autocorrelation

Syntax

```
varlmar [ , options ]
```

options	Description
<code>m^{lag}(#)</code>	use # for the maximum order of autocorrelation; default is <code>m^{lag}(2)</code>
<code>estimates(<i>estname</i>)</code>	use previously stored results <i>estname</i> ; default is to use active results
<code>separator(#)</code>	draw separator line after every # rows

`varlmar` can be used only after `var` or `svar`; see [TS] [var](#) and [TS] [var svar](#).
You must `tsset` your data before using `varlmar`; see [TS] [tsset](#).
`collect` is allowed; see [U] [11.1.10 Prefix commands](#).

Options

- `mlag(#)` specifies the maximum order of autocorrelation to be tested. The integer specified in `mlag()` must be greater than 0; the default is 2.
- `estimates(estname)` requests that `varlmar` use the previously obtained set of `var` or `svar` estimates stored as *estname*. By default, `varlmar` uses the active results. See [R] [estimates](#) for information on manipulating estimation results.
- `separator(#)` specifies how often separator lines should be drawn between rows. By default, separator lines do not appear. For example, `separator(1)` would draw a line between each row, `separator(2)` between every other row, and so on.

Remarks and examples

[stata.com](#)

Most postestimation analyses of VAR models and SVAR models assume that the disturbances are not autocorrelated. `varlmar` implements the LM test for autocorrelation in the residuals of a VAR model discussed in [Johansen \(1995, 21–22\)](#). The test is performed at lags $j = 1, \dots, mlag()$. For each j , the null hypothesis of the test is that there is no autocorrelation at lag j .

`varlmar` uses the estimation results stored by `var` or `svar`. By default, `varlmar` uses the active estimation results. However, `varlmar` can use any previously stored `var` or `svar` estimation results specified in the `estimates()` option.

► Example 1: After var

Here we refit the model with German data described in [TS] [var](#) and then call `varlmar`.

```
. use https://www.stata-press.com/data/r18/lutkepohl2
(Quarterly SA West German macro data, Bil DM, from Lutkepohl 1993 Table E.1)
. var dln_inv dln_inc dln_consump if qtr<=tq(1978q4), dfk
(output omitted)
. varlmar, mlag(5)
```

Lagrange-multiplier test

lag	chi2	df	Prob > chi2
1	5.5871	9	0.78043
2	6.3189	9	0.70763
3	8.4022	9	0.49418
4	11.8742	9	0.22049
5	5.2914	9	0.80821

H0: no autocorrelation at lag order

Because we cannot reject the null hypothesis that there is no autocorrelation in the residuals for any of the five orders tested, this test gives no hint of model misspecification. Although we fit the VAR with the `dfk` option to be consistent with the example in [TS] [var](#), `varlmar` always uses the ML estimator of Σ . The results obtained from `varlmar` are the same whether or not `dfk` is specified.

◀

► Example 2: After svar

When `varlmar` is applied to estimation results produced by `svar`, the sequence of LM tests is applied to the underlying VAR. See [TS] [var svar](#) for a description of how an SVAR model builds on a VAR. In this example, we fit an SVAR that has an underlying VAR with two lags that is identical to the one fit in the previous example.

```
. matrix A = (.,.,0\0,.,.,0\.,.,.)
. matrix B = I(3)
. svar dln_inv dln_inc dln_consump if qtr<=tq(1978q4), dfk aeq(A) beq(B)
(output omitted)
. varlmar, mlag(5)
```

Lagrange-multiplier test

lag	chi2	df	Prob > chi2
1	5.5871	9	0.78043
2	6.3189	9	0.70763
3	8.4022	9	0.49418
4	11.8742	9	0.22049
5	5.2914	9	0.80821

H0: no autocorrelation at lag order

Because the underlying VAR(2) is the same as the previous example (we assure you that this is true), the output from `varlmar` is also the same.

◀

Stored results

`varlmar` stores the following in `r()`:

Matrices

`r(lm)` χ^2 , df, and p -values

Methods and formulas

The formula for the LM test statistic at lag j is

$$\text{LM}_s = (T - d - 0.5) \ln \left(\frac{|\hat{\Sigma}|}{|\tilde{\Sigma}_s|} \right)$$

where T is the number of observations in the VAR; d is explained below; $\hat{\Sigma}$ is the maximum likelihood estimate of Σ , the variance–covariance matrix of the disturbances from the VAR; and $\tilde{\Sigma}_s$ is the maximum likelihood estimate of Σ from the following augmented VAR.

If there are K equations in the VAR, we can define \mathbf{e}_t to be a $K \times 1$ vector of residuals. After we create the K new variables $\mathbf{e}1$, $\mathbf{e}2$, \dots , $\mathbf{e}K$ containing the residuals from the K equations, we can augment the original VAR with lags of these K new variables. For each lag s , we form an augmented regression in which the new residual variables are lagged s times. Per the method of [Davidson and MacKinnon \(1993, 358\)](#), the missing values from these s lags are replaced with zeros. $\tilde{\Sigma}_s$ is the maximum likelihood estimate of Σ from this augmented VAR, and d is the number of coefficients estimated in the augmented VAR. See [\[TS\] var](#) for a discussion of the maximum likelihood estimate of Σ in a VAR.

The asymptotic distribution of LM_s is χ^2 with K^2 degrees of freedom.

References

- Davidson, R., and J. G. MacKinnon. 1993. *Estimation and Inference in Econometrics*. New York: Oxford University Press.
- Johansen, S. 1995. *Likelihood-Based Inference in Cointegrated Vector Autoregressive Models*. Oxford: Oxford University Press.

Also see

- [\[TS\] var](#) — Vector autoregressive models
- [\[TS\] var intro](#) — Introduction to vector autoregressive models
- [\[TS\] var svar](#) — Structural vector autoregressive models
- [\[TS\] varbasic](#) — Fit a simple VAR and graph IRFs or FEVDs

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.

